

## NAG Library Routine Document

### F07PAF (DSPSV)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

## 1 Purpose

F07PAF (DSPSV) computes the solution to a real system of linear equations

$$AX = B,$$

where  $A$  is an  $n$  by  $n$  symmetric matrix stored in packed format and  $X$  and  $B$  are  $n$  by  $r$  matrices.

## 2 Specification

```
SUBROUTINE F07PAF (UPLO, N, NRHS, AP, IPIV, B, LDB, INFO)
  INTEGER          N, NRHS, IPIV(N), LDB, INFO
  REAL (KIND=nag_wp) AP(*), B(LDB,*)
  CHARACTER(1)    UPLO
```

The routine may be called by its LAPACK name *dspsv*.

## 3 Description

F07PAF (DSPSV) uses the diagonal pivoting method to factor  $A$  as  $A = UDU^T$  if UPLO = 'U' or  $A = LDL^T$  if UPLO = 'L', where  $U$  (or  $L$ ) is a product of permutation and unit upper (lower) triangular matrices,  $D$  is symmetric and block diagonal with 1 by 1 and 2 by 2 diagonal blocks. The factored form of  $A$  is then used to solve the system of equations  $AX = B$ .

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

## 5 Arguments

- 1: UPLO – CHARACTER(1) *Input*  
*On entry:* if UPLO = 'U', the upper triangle of  $A$  is stored.  
 If UPLO = 'L', the lower triangle of  $A$  is stored.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of linear equations, i.e., the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .

- 3: NRHS – INTEGER *Input*  
*On entry:*  $r$ , the number of right-hand sides, i.e., the number of columns of the matrix  $B$ .  
*Constraint:*  $\text{NRHS} \geq 0$ .
- 4: AP(\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array AP must be at least  $\max(1, N \times (N + 1)/2)$ .  
*On entry:* the  $n$  by  $n$  symmetric matrix  $A$ , packed by columns.  
 More precisely,  
   if UPLO = 'U', the upper triangle of  $A$  must be stored with element  $A_{ij}$  in  
   AP( $i + j(j - 1)/2$ ) for  $i \leq j$ ;  
   if UPLO = 'L', the lower triangle of  $A$  must be stored with element  $A_{ij}$  in  
   AP( $i + (2n - j)(j - 1)/2$ ) for  $i \geq j$ .  
*On exit:* the block diagonal matrix  $D$  and the multipliers used to obtain the factor  $U$  or  $L$  from  
 the factorization  $A = UDU^T$  or  $A = LDL^T$  as computed by F07PDF (DSPTRF), stored as a  
 packed triangular matrix in the same storage format as  $A$ .
- 5: IPIV(N) – INTEGER array *Output*  
*On exit:* details of the interchanges and the block structure of  $D$ . More precisely,  
   if IPIV( $i$ ) =  $k > 0$ ,  $d_{ii}$  is a 1 by 1 pivot block and the  $i$ th row and column of  $A$  were  
   interchanged with the  $k$ th row and column;  
   if UPLO = 'U' and IPIV( $i - 1$ ) = IPIV( $i$ ) =  $-l < 0$ ,  $\begin{pmatrix} d_{i-1,i-1} & \bar{d}_{i,i-1} \\ \bar{d}_{i,i-1} & d_{ii} \end{pmatrix}$  is a 2 by 2 pivot  
   block and the ( $i - 1$ )th row and column of  $A$  were interchanged with the  $l$ th row and  
   column;  
   if UPLO = 'L' and IPIV( $i$ ) = IPIV( $i + 1$ ) =  $-m < 0$ ,  $\begin{pmatrix} d_{ii} & d_{i+1,i} \\ d_{i+1,i} & d_{i+1,i+1} \end{pmatrix}$  is a 2 by 2 pivot  
   block and the ( $i + 1$ )th row and column of  $A$  were interchanged with the  $m$ th row and  
   column.
- 6: B(LDB,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array B must be at least  $\max(1, \text{NRHS})$ .  
*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ .  
*On exit:* if INFO = 0, the  $n$  by  $r$  solution matrix  $X$ .
- 7: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F07PAF  
 (DSPSV) is called.  
*Constraint:*  $\text{LDB} \geq \max(1, N)$ .
- 8: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element  $\langle value \rangle$  of the diagonal is exactly zero. The factorization has been completed, but the block diagonal matrix  $D$  is exactly singular, so the solution could not be computed.

## 7 Accuracy

The computed solution for a single right-hand side,  $\hat{x}$ , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and  $\epsilon$  is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where  $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$ , the condition number of  $A$  with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) and Chapter 11 of Higham (2002) for further details.

F07PBF (DSPSVX) is a comprehensive LAPACK driver that returns forward and backward error bounds and an estimate of the condition number. Alternatively, F04BJF solves  $AX = B$  and returns a forward error bound and condition estimate. F04BJF calls F07PAF (DSPSV) to solve the equations.

## 8 Parallelism and Performance

F07PAF (DSPSV) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately  $\frac{1}{3}n^3 + 2n^2r$ , where  $r$  is the number of right-hand sides.

The complex analogues of F07PAF (DSPSV) are F07PNF (ZHPSV) for Hermitian matrices, and F07QNF (ZSPSV) for symmetric matrices.

## 10 Example

This example solves the equations

$$Ax = b,$$

where  $A$  is the symmetric matrix

$$A = \begin{pmatrix} -1.81 & 2.06 & 0.63 & -1.15 \\ 2.06 & 1.15 & 1.87 & 4.20 \\ 0.63 & 1.87 & -0.21 & 3.87 \\ -1.15 & 4.20 & 3.87 & 2.07 \end{pmatrix} \quad \text{and} \quad b = \begin{pmatrix} 0.96 \\ 6.07 \\ 8.38 \\ 9.50 \end{pmatrix}.$$

Details of the factorization of  $A$  are also output.

## 10.1 Program Text

```

Program f07pafe

!      F07PAF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: dspsv, nag_wp, x04ccf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
Character (1), Parameter   :: uplo = 'U'
!      .. Local Scalars ..
Integer                    :: i, ifail, info, j, n
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: ap(:), b(:)
Integer, Allocatable       :: ipiv(:)
!      .. Executable Statements ..
Write (nout,*) 'F07PAF Example Program Results'
Write (nout,*)
Skip heading in data file
Read (nin,*)
Read (nin,*) n

Allocate (ap((n*(n+1))/2),b(n),ipiv(n))

!      Read the upper or lower triangular part of the matrix A from
!      data file

If (uplo=='U') Then
  Read (nin,*)((ap(i+(j*(j-1))/2),j=i,n),i=1,n)
Else If (uplo=='L') Then
  Read (nin,*)((ap(i+((2*n-j)*(j-1))/2),j=1,i),i=1,n)
End If

!      Read b from data file

Read (nin,*) b(1:n)

!      Solve the equations Ax = b for x
!      The NAG name equivalent of dspsv is f07paf
Call dspsv(uplo,n,1,ap,ipiv,b,n,info)

If (info==0) Then

!      Print solution

Write (nout,*) 'Solution'
Write (nout,99999) b(1:n)

!      Print details of factorization

Write (nout,*)
Flush (nout)

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04ccf(uplo,'Non-unit diagonal',n,ap,
'Details of the factorization',ifail)

!      Print pivot indices

Write (nout,*)
Write (nout,*) 'Pivot indices'
Write (nout,99998) ipiv(1:n)

Else

```

```

        Write (nout,99997) 'The diagonal block ', info, ' of D is zero'
      End If

99999 Format ((3X,7F11.4))
99998 Format (1X,7I11)
99997 Format (1X,A,I3,A)
      End Program f07pafe

```

## 10.2 Program Data

```

F07PAF Example Program Data
  4                               :Value of N
-1.81   2.06   0.63  -1.15
         1.15   1.87   4.20
                -0.21   3.87
                        2.07 :End of matrix A
  0.96   6.07   8.38   9.50 :End of vector b

```

## 10.3 Program Results

```

F07PAF Example Program Results

Solution
  -5.0000   -2.0000    1.0000    4.0000

Details of the factorization
      1      2      3      4
1      0.4074    0.3031  -0.5960    0.6537
2              -2.5907    0.8115    0.2230
3                      1.1500    4.2000
4                              2.0700

Pivot indices
      1      2      -2      -2

```

---